问题 6 料场的建立与运输

建筑工地的位置(用平面坐标 a，b 表示，距离单位：公里)及水泥日用量 d(吨)下表给出。有两个临时料场位于 P(5,1)，Q(2,7)，日储量各有 20 吨。
(1)从 P,Q 两料场分别向各工地运送多少吨水泥，使总的吨公里数最小。
(2)两个新的料场应建在何处，节省的吨公里数有多大？

表 1 运输表

	1	2	3	4	5	6
x	1.25	8.75	0.5	5.75	3	7.25
y	1.25	0.75	4.75	5	6.5	7.75
d	3	5	4	7	6	11

1. 符号说明

表 2 符号说明

符号	说明
t_{ij}	第 j 个供水泥的地区对第 i 个地区的供水泥量
$x1_i$	需要水泥的地区（建筑工地）中第 i 个地区的横坐标
$y1_i$	需要水泥的地区（建筑工地）中第 i 个地区的纵坐标
$x2_j$	提供水泥的地区（料场）中第 j 个地区的横坐标
$y2_j$	提供水泥的地区（料场）中第 j 个地区的纵坐标
d_i	第 i 个地区对水泥的需求量
s_j	第 j 个供水泥地区的水泥存储量

在第（1）问中

P(5,1) = P($x2_1$，$y2_1$)

Q(2,7) = Q($x2_2$，$y2_2$)

2. 问题分析

约束 1：水泥储量，水泥需求量，同时水泥的运输量要大于 0；
约束 2：根据水泥需求量，每需求地的水泥供量不得低于其需求；
约束 3：根据水泥储量，每地的水泥不多于 20 吨；

3. 模型建立

目标函数：二维坐标，料场和工地之间相连道路为直线

$$minD = \sum_{i=1}^{6}\sum_{j=2}^{2} t_{ij}\sqrt{(x1_i - x2_j)^2 + (y1_i - y2_j)^2}$$

根据水泥需求量，每需求地的水泥供量不得低于其需求：

$$\sum_{j=1}^{2} t_{ij} \geq d_i \quad i = 1,2,3,4,5,6$$

根据水泥储量，每地的水泥不多于 20 吨:

$$\sum_{i=1}^{6} t_{ij} \le s_j \quad j = 1,2$$

水泥的运输量要大于 0:

$$t_{ij} \ge 0, i = 1,2,3,4,5,6, j = 1,2$$

数学模型:

$$minD = \sum_{i=1}^{6}\sum_{j=2}^{2} t_{ij}\sqrt{(x1_i - x2_j)^2 + (y1_i - y2_j)^2}$$

$$s.t = \begin{cases} \sum_{j=1}^{2} t_{ij} \ge d_i \quad i = 1,2,3,4,5,6 \\ \sum_{i=1}^{6} t_{ij} \le s_j \quad j = 1,2 \\ t_{ij} \ge 0 \end{cases}$$

4. 结果分析

4.1 从 P, Q 两料场分别向各工地运送多少吨水泥，使总的吨公里数最小

表 3 运输方案

	1	2	3	4	5	6
P	3	5	0	7	0	1
Q	0	0	4	0	6	10

总的吨公里数最小最少为 136.2 吨公里

Lingo 代码:

```
model:
sets:
row/1..6/:x,y,d;
col/1..2/;
pp(col,row):ans;
endsets
data:
d=3,5,4,7,6,11;
x=1.25,8.75,0.5,5.75,3,7.25;
y=1.25,0.75,4.75,5,6.5,7.75;
enddata
min=@sum(row(i):@sqrt((x(i)-5)^2+(y(i)-1)^2)*ans(1,i)+@sqrt((x(i)-
2)^2+(y(i)-7)^2)*ans(2,i));
@sum(row(i):ans(1,i))<=20;
@sum(row(i):ans(2,i))<=20;
@for(row(i):ans(1,i)+ans(2,i)>=d(i));

end
```

4.2 两个新的料场应建在何处，节省的吨公里数有多大

总的吨公里数 Objective value 最小为 89.88347

两个新的料场坐标

$(x2_1, \ y2_1)$=(5. 695966, 4. 928558)

$(x2_2, \ y2_2)$=(7. 25, 7. 75)

Lingo 代码:

```
model:
sets:
kc/1..2/:x,y,e;
gd/1..6/:a,b,d;
link(kc,gd):c;
endsets
data:
a=1.25,8.75,0.5,5.75,3,7.25;
b=1.25,0.75,4.75,5,6.5,7.75;
d=3,5,4,7,6,11; e=20,20;
enddata
init:
x,y=5,1,2,7;
endinit
min=@sum(link(i,j):c(i,j)*((x(i)-a(j))^2+(y(i)-b(j))^2)^(1/2));
@for(kc(i):@sum(gd(j):c(i,j))<e(i));
@for(gd(j):@sum(kc(i):c(i,j))>d(j));
@for(kc:@bnd(0.5,x,8.75));
@for(kc:@bnd(0.75,y,7.75));

end
```

问题 7 模糊规划

某药品加工厂生产甲乙两种药品，甲种药品每千克利润 3 万元，乙种药品每千克利润 4 万元。生产每千克甲种药品需要原料 A 略少于 4kg，需要原料 B 约 12kg。生产每千克乙种药品需要原料 A 略多于 20kg，需要原料 B 约 6.4kg。现原料 A 还有约 4600kg，原料 B 还有约 4800kg。如何安排甲乙两种药品的产量以使利润最大。

表 1 药品材料表

	原料 A	原料 B	
甲药品	4	12	3
乙药品	20	6.4	4
	4600	4800	

1.符号说明

表 2 符号说明

符号	说明
$\boldsymbol{x_1}$	甲种药品的产量
$\boldsymbol{x_2}$	乙种药品的产量
$\boldsymbol{s_1}$	原料 A 的存储量
$\boldsymbol{s_2}$	原料 B 的存储量
$\boldsymbol{d_1}$	对原料 A 需求的伸缩指标
$\boldsymbol{d_2}$	对原料 B 需求的伸缩指标
$\boldsymbol{\lambda}$	隶属度

2.问题分析

约束 1：现有原料 A 约 4600kg，原料 B 约 4800kg；

约束 2：甲乙两种药品的产量大于 0；

约束 3：隶属度λ在 0 到 1 之间

3.模型建立

目标函数：生产甲乙两种药品利润最大

$$maxf=3x_1+4x_2$$

约束 1: 现有原料 A 约 4600kg，原料 B 约 4800kg:

$$4x_1+20x_2\le[s_1,d_1]$$

$$12x_1+6.4x_2\le[s_2,d_2]$$

约束条件 3：两种药品的产量都是非负数

$$x_1,x_2\ge 0$$

约束条件 4：隶属度的范围在 0 到 1 之间

$$\lambda\in[0,1]$$

数学模型:

$$maxf=3x_1+4x_2$$

$$s.t=\begin{cases}4x_1+20x_2\le[s_1,d_1]\\12x_1+6.4x_2\le[s_2,d_2]\\x_1,x_2\ge 0\\\lambda\in[0,1]\end{cases}$$

3.1 普通线性规划

$$maxf_1=3x_1+4\boldsymbol{x_2}$$

$$s.t=\begin{cases}4x_1+20x_2\le s_1\\12x_1+6.4x_2\le s_2\\x_1,x_2\ge 0\\\lambda\in[0,1]\end{cases}$$

3.2 添加伸缩指标

$$maxf_2=3x_1+4\boldsymbol{x_2}$$

$$s.t=\begin{cases}s_1-d_1\le 4x_1+20x_2\le s_1+d_1\\s_2-d_2\le 12x_1+6.4x_2\le s_2+d_2\\x_1,x_2\ge 0\\\lambda\in[0,1]\end{cases}$$

3.3 添加新的变量隶属度 λ

$$max\lambda$$

$$s.t=\begin{cases}3x+4y-(f_2-f_1)\lambda\ge f_1\\4x_1+20x_2+d_1\lambda\le s_1+d_1\\4x_1+20x_2-d_1\lambda\ge s_1-d_1\\12x_1+6.4x_2+d_2\lambda\le s_2+d_2\\12x_1+6.4x_2-d_2\lambda\ge s_2-d_2\\x_1,x_2\ge 0\\\lambda\in[0,1]\end{cases}$$

4.结果分析

表 3 结果表

	x_1	x_2	max	λ
普通线性规划	310.4478	167.9104	1602.985	
增加伸缩指标	321.4552	170.7090	1647.201	
增加隶属度	315.9515	169.3097	1625.0933	0.5000032

Lingo 代码:

普通线性规划

```
model:
data:
b1=4600;
b2=4800;
enddata
max=3*x+4*y;
4*x+20*y<b1;
12*x+6.4*y<b2;
x>0;
y>0;
```

增加伸缩指标的线性规划:

```
model:
data:
b1=4600;
b2=4800;
d1=100;
d2=150;
enddata
max=3*x+4*y;
4*x+20*y<b1+d1;
4*x+20*y>b1-d1;
12*x+6.4*y<b2+d2;
12*x+6.4*y>b2-d2;
```

```
x>0;
y>0;
```

增加隶属度的线性规划：

```
model:
data:
b1=4600;
b2=4800;
d1=100;
d2=150;
f1=1602.985;
f2=1647.201;
enddata
max=lamada;
3*x+4*y-(f2-f1)*lamada>f1;
4*x+20*y+d1*lamada<b1+d1;
4*x+20*y-d1*lamada>b1-d1;
12*x+6.4*y+d2*lamada<b2+d2;
12*x+6.4*y-d2*lamada>b2-d2;
x>0;
y>0;
lamada>0;
lamada<1;
```

问题 8 下料问题

某钢管零售商从钢管厂进货，将钢管按照顾客的要求切割后售出。从钢管厂进货时得到的原料钢管长度都是 1850㎜。现有一位客户需要 15 根 290㎜、28 根 315㎜、21 根 350㎜ 和 30 根 455㎜ 的钢管。为了简化生产过程，规定所使用的切割模式的种类不能超过 4 种，使用频率最高的一种切割模式按照一根原料钢管价值的 1/10 给予加工费用，使用频率次之的切割模式按照一根原料钢管价值的 2/10 给予加工费用，依此类推，且每种切割模式下的切割次数不能太多（一根原料钢管最多生产 5 根产品）。此外，为了减少余料浪费，每种切割模式下的余料浪费不能超过 100㎜。为了使总费用最小，应如何下料？

表 1 四种切割模式下料方案

	290mm	315mm	350mm	455mm
1				
2				
3				
4				
	15	28	21	30

1. 符号说明

表 2 符号说明

符号	说明
x_{ij}	一根原料在第 i 种模式下切第 j 种钢管的数量
n_i	第 i 种模式需要的原料钢管数目
d_j	第 j 种钢管的需要量
l_j	第 j 种钢管的长度
b	原料钢管的长度
c_i	使用第 i 种模式每根钢管的费用
f	一根钢管被切割的最大次数
e	最大余料长度

2. 问题分析

约束 1：规定所使用的切割模式的种类不能超过 4 种，采用 4 种类切割模式节省下料总费用；

约束 2：每种模式切割下的顾客需求的钢管的总数要大于顾客的需求；

约束 3：一根原料钢管最多生产 5 根产品；

约束 4：每种切割模式下的余料浪费不能超过 100mm；

约束 5：使用频率最高的一种切割模式按照一根原料钢管价值的 1/10 给予加工费用，使用频率次之的切割模式按照一根原料钢管价值的 2/10 给予加工费用，依此类推。切割模式使用频率依次递减；

约束 6：切割所用的原料钢管的和以及切割下来的需求钢管的数目都是整数；

约束 7：在原料钢管没有剩余的前提下原料钢管的总数要大于不考虑任何限制下最小的需求量

3. 模型建立

目标函数，希望总费用最小：

$$min\,f=\sum_{i=1}^{4}n_ic_i$$

（1）约束 1：一根原料钢管在每种切割模式下的余料浪费不大于 100mm。

$$\sum_{j=1}^{4}x_{ij}l_j\geq b-e，i=1,2,3,4$$

（2）约束 2：一根原料钢管最多生产 5 根产品。

$$\sum_{j=1}^{4}x_{ij}\leq f，i=1,2,3,4$$

（3）约束 3：每种模式切出的钢管总长度不大于原料钢管长度。

$$\sum_{j=1}^{4}x_{ij}l_j\leq b，i=1,2,3,4$$

（4）约束 4：切出的每种钢管总数量不小于顾客的需求。

$$\sum_{i=1}^{4}x_{ij}n_i\geq d_j\,,j=1,2,3,4$$

（5）约束 5：每种切割模式生产产品的根数及每种切割模式所需钢管原料的根数为整数。

$$x_{ij},n_i\in Z，i=1,2,3,4\;j=1,2,3,4$$

根据以上目标函数及约束条件，建立以下模型：

$$min\,f=\sum_{i=1}^{4}n_ic_i$$

$$s.t.=\begin{cases}\sum_{j=1}^{4}x_{ij}l_i\geq b-e,i=1,2,3,4\\ \sum_{j=1}^{4}x_{ij}\leq f,i=1,2,3,4\\ \sum_{j=1}^{4}x_{ij}l_i\leq b,i=1,2,3,4\\ \sum_{i=1}^{4}x_{ij}a_i\geq d_j,j=1,2,3,4\\ x_{ij},a_i\in Z，i=1,2,3,4\;j=1,2,3,4\end{cases}$$

4. 结果分析

表 3 各切割模式下各产品的切割数量

切割模式	290mm	315mm	350mm	455mm	总根数
模式 1	1	2	0	2	5

模式 2	0	0	5	0	5
模式 3	2	0	1	2	5
模式 4	2	1	0	2	5

使得所需的总费用最少，最少为 215 元

Lingo 代码：

```
model:
sets:
ms/1..4/:x,p;
xq/1..4/:d,a;
link(xq,ms):r;
endsets
data:
d=15 28 21 30;
L=1850;
e=100;
k=5;
p=1.1 1.2 1.3 1.4;
a=290 315 350 455;
enddata
min=@sum(ms(j):x(j)*p(j));
@for(xq(i):@sum(ms(j):x(j)*r(i,j))>d(i));
@for(ms(j):@sum(xq(i):a(i)*r(i,j))<L);
@for(ms(j):@sum(xq(i):a(i)*r(i,j))>L-e);
@for(ms(j):@sum(xq(i):r(i,j))<k);
@for(ms(j)|j#lt#@size(ms):x(j)>x(j+1));
@for(ms:@gin(x));
@for(link:@gin(r));
@sum(ms(j):x)>@floor(@sum(xq:a*d)/L)+1;
```